
Trilingual interactive Table of Mendeleev  
on HP Prime 

 
Elem_Case(n,s,a) 
//Affichage de la case d'un élément (en blanc si s=1) 
BEGIN 
LOCAL sb,lg,cl,bl=#FFFFFF,coul; 
 
sb:=IFTE(a,Mn_Cl.Cell(20,3),Mndlv.Cell(n,1)); 
lg:=23*Mndlv.Cell(n,14)-15; 
cl:=17*Mndlv.Cell(n,13)-7; 
coul:=Mndlv.Cell(n,12); 
 
IF s>0 THEN 
  RECT_P(cl,lg,cl+15,lg+21,0,bl); 
ELSE 
  RECT_P(cl,lg,cl+15,lg+21,Mn_Cl.Cell(coul,1)); 
END; 
TEXTOUT_P(sb,cl+1,lg+2,1); 
 
END; 
 
 
 
Elem_Car(n) 
//Affichage des caractéristiques d'un élément 
BEGIN 
LOCAL coul,ic,lb,gd,un,nu; 
 
coul:=Mndlv.Cell(n,12); 
RECT_P(47,1,212,75);RECT_P(130,2,272,13); 
TEXTOUT_P(Mndlv.Cell(n,1+D),47,2,2,Mn_Cl.Cell(coul,1)); 
TEXTOUT_P(Mn_Cl.Cell(coul,1+D),135,4,1,Mn_Cl.Cell(coul,1)); 
 
FOR ic FROM 1 TO 6 DO 
    lb:=Mn_Cl.Cell(ic+10,1+D); 
    gd:=Mndlv.Cell(n,ic+4); 
    un:=Mn_Cl.Cell(ic+10,1); 
    nu:=Mn_Cl.Cell(ic+10,4+D);      
    TEXTOUT_P(lb+IFTE(gd,gd+un,nu),47,10*ic+5,1); 
END; 
 
END; 
 
 
EXPORT Mndl_Dsp() 
//Affichage de la table de Mendeleiev 
BEGIN 
LOCAL ie,na,lst,aut; 
 
na:=LEFT(Mn_Cl.Cell(20,2),11); 
lst:=ASC(MID(na,5,1)); 
A:=1-(lst(1)=233); 
 
RECT_P(); 
FOR ie FROM 1 TO 112 DO 
    Elem_Case(ie,0,A); 
END; 
 
TEXTOUT_P(na,235,180,1,Mn_Cl.Cell(20,1)); 
RECT_P(47,169,49,225,0); 
RECT_P(49,200,60,202,0); 
RECT_P(49,224,60,226,0); 
 
END;

EXPORT Mendeleiev() 
//Navigation dans le tableau de Mendeleiev 
BEGIN 
LOCAL nv,ie,n_old,n,nn,lgn=5,cln=11,nk; 
LOCAL an,titre,list_choix,ic,lst_alph,ia,raton,tmpo=0.2; 
L9:={14,15,16,17,18,20,21,22,23,24,25,26,27,28,29,31,32,33,34,35,37,38,39,40,42,43}; 
 
// choix de la langue D=1 en Français, D=2 in English, D=3 en Español 
titre:=Mn_Cl.Cell(17,2); 
list_choix:={Mn_Cl.Cell(18,2),Mn_Cl.Cell(18,3),Mn_Cl.Cell(18,4)}; 
D:=1;CHOOSE(D,titre,list_choix); 
titre:=Mn_Cl.Cell(17,1+D); 
 
REPEAT 
  list_choix:={Mn_Cl.Cell(19,1+D),Mn_Cl.Cell(19,4+D)+"A, B, C,..."}; 
  // si nv=1, navigation 2D touches et souris, si nv=2, navigation alpha 
  CHOOSE(nv,titre,list_choix); 
  IF nv=2 THEN 
     lst_alph:=MAKELIST(" ",X,1,112); 
     FOR ie FROM 1 TO 112 DO 
         lst_alph(ie):=LEFT(Mndlv.Cell(Mndlv.Cell(ie,14+D),1+D),1); 
     END;          
  END; 
  // on affiche la table 
  Mndl_Dsp(); 
 
  n_old:=IFTE(nv=1,47,Mndlv.Cell(1,14+D));n:=n_old;ie:=1; 
  WHILE nv DO 
    // on pointe sur le "premier" élément 
    Elem_Case(n,1,A); 
    RECT_P(5,200,42,236); 
    TEXTOUT_P(IFTE(D=1,"an",IFTE(D=2,"year","año")),6,200,1); 
    an:=Mndlv.Cell(n,11); 
    IF an THEN TEXTOUT_P(an,6,210,2) END; 
    TEXTOUT_P("n="+n,6,225,1); 
 
    // on affiche les données de l'élément courant 
    Elem_Car(n); 
 
    // boucle de navigation mixte (touches/souris) dans le tableau 
    //____________________________________________________________  
    WHILE 1 DO 
       // si la souris est autorisée (navigation 2D) 
       IF nv==1 THEN 
          raton:=MOUSE(); 
          IF SIZE(raton(1)) THEN 
             // si on a cliqué avec la souris dans une case 
             cln:=FLOOR((IP(raton(1,1)-10))/17)+1;            
             lgn:=FLOOR((IP(raton(1,2)-8))/23)+1; 
             IF lgn>O AND cln>0 THEN 
                IF lgn<=7 THEN 
                   nn:=Mn_Nv.Cell(lgn,cln); 
                   IF nn>0 THEN  
                      n_old:=n;Elem_Case(n_old,0,A); 
                      n:=nn;BREAK; 
                   END; 
                END; 
                IF lgn>8  THEN 
                   lgn:=lgn-1;           
                   nn:=Mn_Nv.Cell(lgn,cln); 
                   IF nn>0 THEN  
                      n_old:=n;Elem_Case(n_old,0,A); 
                      n:=nn;BREAK; 
                   END; 
                END; 
             END; 
          END; 
       END; 



       // si on a touché au clavier (flèches) 
       CASE 
           IF ISKEYDOWN(4) THEN BREAK(2); END; 
           IF ISKEYDOWN(2) AND lgn>0 AND nv==1 THEN 
              nn:=Mn_Nv.Cell(lgn-1,cln); 
              IF nn>0 THEN 
                 n_old:=n;Elem_Case(n_old,0,A); 
                 lgn:=lgn-1;n:=nn; 
              END; 
              WAIT(tmpo);BREAK;  
           END; 
           IF ISKEYDOWN(2) AND ie>1 AND nv==2 THEN 
              n_old:=n;Elem_Case(n_old,0,A); 
              ie:=ie-1;n:=Mndlv.Cell(ie,14+D); 
              WAIT(tmpo);BREAK; 
           END; 
           IF ISKEYDOWN(12) AND nv==1 THEN 
              nn:=Mn_Nv.Cell(lgn+1,cln); 
              IF nn>0 THEN 
                 n_old:=n;Elem_Case(n_old,0,A); 
                 lgn:=lgn+1;n:=nn; 
              END; 
              WAIT(tmpo);BREAK; 
           END; 
           IF ISKEYDOWN(12) AND ie<112 AND nv==2 THEN 
              n_old:=n;Elem_Case(n_old,0,A); 
              ie:=ie+1;n:=Mndlv.Cell(ie,14+D); 
              WAIT(tmpo);BREAK; 
           END; 
           IF ISKEYDOWN(7) AND cln>0 AND nv==1 THEN 
              nn:=Mn_Nv.Cell(lgn,cln-1); 
              IF nn>0 THEN 
                 n_old:=n;Elem_Case(n_old,0,A); 
                 cln:=cln-1;n:=nn; 
              END; 
              WAIT(tmpo);BREAK; 
           END; 
           IF ISKEYDOWN(8) AND nv==1 THEN 
              nn:=Mn_Nv.Cell(lgn,cln+1); 
              IF nn>0 THEN 
                 n_old:=n;Elem_Case(n_old,0,A); 
                 cln:=cln+1;n:=nn; 
              END; 
              WAIT(tmpo);BREAK; 
           END; 
       END; 
       // touches A,B,C, ... du clavier (navigation alpha) 
       IF nv==2 THEN  
          ic:=1; 
          REPEAT                       
             IF ISKEYDOWN(L9(ic)) THEN 
                ia:=POS(lst_alph,CHAR(64+ic)); 
                IF ia THEN  
                   n_old:=n;Elem_Case(n_old,0,A); 
                   ie:=ia;n:=Mndlv.Cell(ie,14+D); 
                   WAIT(tmpo);BREAK(2); 
                END 
             END; 
             ic:=ic+1 
          UNTIL ic>26; 
       END   
 
    END; 
    //____________________________________________________________  
  END 
UNTIL NOT nv; 
 
END; 

Screen captures of the HP Prime (trilingual software: English, Français and Español) 
 

 
 

Touch (or mouse) navigation and Keyboard navigation (=,\,<,>, A, B, C,…) 
 
Data are stored in Spreadsheets created and named with the Spreadsheet App of the HP Prime : 
 

 The characteristics of the elements of the Periodic Table are stored in Mndlv spreadsheet. 
 Types and Colors of categories of elements and trilingual menus are stored in Mn_Cl spreadsheet. 
 The locations of the elements in the periodic table are in Mn_Nv spreadsheet. 

 
Thus it is possible to modify the data (or to translate some) without interfering with interactive navigation 
algorithm stored in the program that is independent of data. 

